
Incidence estimates are used to measure trends in the initiation of substance use using the National Survey on Drug Use and Health (NSDUH) data. Two general types of incidence estimates can be generated using NSDUH data: calendar year incidence and past year incidence. Additionally, for respondents defined as having used illicit drugs for the first time within the past 12 months of their interview date, the drug that was initiated first can also be determined. Measures of calendar and past year incidence are different in their conceptualization, calculation, and interpretation. Examples showing how to compute past year initiation estimates, mean age of first use, and past year initiate drug of first use are also included at the end of this appendix.
Incidence
"Calendar year initiates" refer to the number of substance users reporting that they first used a substance within each prior calendar year (between January 1 and December 31), based on age-at-first-use data collected from all persons who have used the substance at least once in his or her lifetime. Together with information on the interview date and date of birth, age at first use is used to calculate the calendar year in which initial use occurred. Tabulated across calendar years, these estimates can indicate long-term incidence trends. (To avoid counting initiation that occurred outside the United States, the date of entry into the United States is taken into account for respondents who indicated they were not born in the United States.) Calendar year incidence was first presented in the 1999 NSDUH and continued through the 2004 NSDUH. An example showing how to create indicator variables and calculate calendar year incidence estimates can be found in the 2004 or earlier NSDUH analytic codebooks. Calendar year incidence estimates for periods earlier in the respondents' lives may be subject to substantial memory-related bias, thus potentially distorting the data for earlier years in long-term trend tables (Gfroerer, Hughes, Chromy, Heller, & Packer, 2004).
Because of concerns about the validity of trend estimates of incidence based on long recall periods, an alternative approach to estimating incidence was developed: past year incidence. This approach reflects initial use in the 12 months leading up to the date when a respondent was interviewed and does not take into account whether the respondent initiated substance use while a resident of the United States. Although this alternative measure will not eliminate reporting biases, the approach should minimize recall bias because its estimates are based on a more recent time period than the previously produced calendar year estimates. The more recent time period also provides more timely data on emerging patterns of use. Another advantage of this approach is that initiation estimates can be analyzed in conjunction with past year prevalence estimates because they reflect the same time period and population of interest (all respondents, not only those initiating in the United States), which provides a more complete and consistent picture of substance users.
Past year incidence was presented for the first time in the 2004 NSDUH and was displayed along with calendar year incidence. However, beginning with the 2005 NSDUH and continuing in subsequent NSDUHs, only past year incidence estimates have been presented; therefore, the 2013 analytic data file contains indicator variables to generate only this type of incidence estimate.
Estimates of the incidence measure, termed "past year initiation," refer to respondents whose date of first use of a substance was within the 12 months prior to their interview date. This measure is determined by self-reported year and month of first use, an imputed day of first use, self-reported past year use (with the exception of daily cigarette use), and the interview date.[footnoteRef:1] [1:  "Self-reported" refers to responses provided by the respondents within the questionnaire. Responses are imputed for respondents who do not self-report for these items. Day-of-first-use data is imputed because this information is not asked in the questionnaire.] 

Past year incidence is reported in the 2013 detailed tables (Center for Behavioral Health Statistics and Quality [CBHSQ], 2014a) and the 2013 summary of national findings (CBHSQ, 2014b) as the number of new users or initiates of a substance in the past 12 months, as well as past year incidence rates (i.e., the number of new users among persons at risk for initiation in the past 12 months). The number of initiates in the past year is estimated by computing the number of persons who began using a given substance during the 12-month period immediately prior to their interview. Past year incidence rates, labeled as the "percentage of past year initiates among persons at risk for initiation" in the report, are calculated as the number of persons initiating substance use in the past 12 months divided by the number of persons eligible to initiate substance use (expressed per 100). The two measures are equivalent to the epidemiological definition of incidence, which is the number of new cases that occur during a specified time period (e.g., 12 months) within a population at risk (e.g., those who have not initiated substance use).
Included at the end of this appendix are examples of how to compute past year initiation measures and age-of-first-use estimates. The past year initiation example A contains SAS® code[footnoteRef:2] to illustrate how to create the numerator and denominator variables that are used to compute cigarette, marijuana, and illicit drug past year incidence estimates (SAS Institute Inc., 2013). The past year initiation example A also includes SUDAAN® code[footnoteRef:3] showing how to create the estimated number of initiates and the rate of initiation (RTI International, 2012), followed by SAS code showing how to invoke the NSDUH suppression criteria. Example B contains SUDAAN code showing how to compute the mean age at first marijuana use among past year marijuana initiates aged 12 to 49 and code for invoking the suppression criteria for averages. For details on the NSDUH suppression criteria, see Table B.2 in Appendix B of the 2013 summary of national findings report (CBHSQ, 2014). Notes have been included within this code to explain the various steps and how similar code can be used to create indicators and estimates for other substances. [2:  SAS® is a registered trademark of SAS Institute Inc.]  [3:  SUDAAN® is a registered trademark of RTI International (a trade name of Research Triangle Institute).] 

For the three substance summary measures (illicit drugs, illicit drugs other than marijuana, and nonmedical use of psychotherapeutics), the date of initiation is defined as the earliest date of initiation among the respective subgroup of substances. The SAS code in the past year initiation example A identifies which substance among the group was most recently initiated and ensures that the past year use indicator variable corresponding to the specific substance with the earliest date of initiation is used in the creation of the past year recent initiation and eligible-for-initiation variables for substance summary measures.
The past year initiation variables are defined separately for recent initiates (RECXXX) and eligible initiates (ELIGXXX). These variables have evolved over time with minor revisions being applied to the definition of these variables, and the impacted variables have been renamed. A summary of the changes has been included below. Given the small impact of these changes, the past year initiation variables can still be considered comparable with 2002 and later NSDUH years. When performing analyses on multiple years of data, users should rename the initiation variables so that the variables have the same name across all years for comparable use in analyses.
Starting with the 2005 NSDUH, the initiation variables were revised to ensure that a respondent was a past year user in order to be classified as an eligible or recent initiate. Before this revision was applied (2002-2004 NSDUHs), a small number of respondents were classified as past year initiates but not past year users. To meet the criterion of a past year initiate, the difference in days between the interview date and the date of first use for a particular substance must be between 0 and 365. Assuming that the day of interview is considered a partial day of "exposure" and interviews are conducted uniformly throughout the day, the time frame actually encompasses approximately 365.5 days of exposure, whereas the time frame would actually be allowing respondents only approximately 364.5 days of exposure (or slightly less than a year). To ensure that all respondents were given a full year of exposure for possible initiation, the slightly longer interval was used. However, because of this decision to use the longer interval (0-365), an inconsistency was discovered with the past year measures. Therefore, an adjustment of adding past year use (XXXYR=1) as a criterion was made. The adjustment provided consistency between the variables that indicate initiation and the variables that report past year use. Given that the number of respondents affected was small, the 2005 and later year initiation variables are considered comparable with the 2002-2004 variables.
For the substance summary measures, the date of initiation is defined as the earliest date of initiation among the respective subgroup of substances. A coding change was implemented starting with the 2007 NSDUH for the summary initiation variables (i.e., SUM, IEM, and PSY for eligible and recent initiates) to ensure that the date of the initiation corresponds to the past year use for the specific drug first used among the group of drugs used to create the summary variable. The addition of the past year criterion in the 2005 NSDUH, discussed above, allowed an additional window for discrepancies for the summary measures between the earliest date of first use and the overall imputed summary measure. For example, a respondent interviewed on 10/6/2007 would have been misclassified as a past year recent initiate of psychotherapeutic drugs because the first reported drug use was pain relievers on 10/6/2006 (exactly 365 days from the interview date), but based on the past year measures the respondent was still classified as a past year user of psychotherapeutics (PSYYR2=1). However, this respondent reported first using stimulants on 9/22/2007. This stimulant use within the past year (less than 365 days from the interview date) should dictate the respondent's classification as a past year psychotherapeutic user (PSYYR2=1), not their non-past year use of pain relievers (exactly 365 days from the interview date). Because this respondent's reported first use of pain relievers was technically 2 years prior, the respondent should not have been considered a past year recent initiate of psychotherapeutics. This change in code starting with the 2007 NSDUH now confirms the past year use status of pain relievers specifically (ANLYR=1) rather than the more broad past year psychotherapeutic drug measure to determine that the respondent did not initiate the use of psychotherapeutics in the past year. Even with this slight change in code, the 2007 and later year initiation variables are considered comparable with the 2002-2004 and the 2005-2006 variables.
Table 1 summarizes the revisions and presents the changes to the eligible and recent initiate variable names across the years.
Table 1.	Recent and Eligible Initiate Variable Comparability: NSDUH, 2002 and Later Years
	Variable
	2002-20041
	2005-20062
	2007 and 
Later Years

	ELIGIBLE INITIATE
	
	
	

	Single Substance Variable
	ELIGXXX3
	ELIGXXX23
	ELIGXXX23

	Daily Cigarettes4
	ELIGCD2
	ELIGCD2
	ELIGCD2

	Summary Variables5
	
	
	

	Illicit Drugs
	ELIGSUM
	ELIGSUM2
	ELIGSUM3

	Illicit Drugs Other Than Marijuana
	ELIGIEM
	ELIGIEM2
	ELIGIEM3

	Nonmedical Psychotherapeutics
	ELIGPSY
	ELIGPSY2
	ELIGPSY3

	RECENT INITIATE
	
	
	

	Single Substance Variable
	RECXXX3
	RECXXX23
	RECXXX23

	Daily Cigarettes4
	RECCD2
	RECCD2
	RECCD2

	Summary Variables5
	
	
	

	Illicit Drugs
	RECSUM
	RECSUM2
	RECSUM3

	Illicit Drugs Other Than Marijuana
	RECIEM
	RECIEM2
	RECIEM3

	Nonmedical Psychotherapeutics
	RECPSY
	RECPSY2
	RECPSY3


NOTE:	All the eligible initiate and recent initiate variables within the same row are comparable across years (i.e., 2002-2004 variables are comparable with 2005-2006 and 2007 and later year variables). Users wanting to analyze across years should rename these variables to match for the years of interest.
1 OxyContin® initiate variables are not available for 2002 and 2003 because OxyContin questions were not included in the questionnaire.
2 Starting with the 2005 NSDUH, all the variables listed in this column (except daily cigarettes) were revised because a new component was added to the criteria for defining these variables. The new criterion required that a respondent indicate past year use (XXXYR=1) to be classified as an eligible or recent initiate. Prior to adding this new criterion, a small number of respondents were classified as past year initiates but not past year users. Given that the number of respondents affected was small, the 2005 and later year initiation variables are considered comparable with the 2002-2004 variables.
3 XXX refers to drug abbreviations such as ALC = Alcohol; ANL = Pain Relievers; CGR = Cigars; CIG = Cigarettes; COC = Cocaine; CRK = Crack; ECS = Ecstasy; HAL = Hallucinogens; HER = Heroin; INH = Inhalants; LSD; MJ = Marijuana and Hashish; MTH = Methamphetamine; OXY = OxyContin; PCP; SED = Sedatives; SLT = Smokeless Tobacco; STM=Stimulants; and TRN = Tranquilizers.
4 The daily cigarette recodes (ELIGCD2 and RECCD2) had no revisions in definition across the years and therefore were not renamed.
5 For substance summary measures (i.e., psychotherapeutics), the date of initiation is defined as the earliest date of initiation among a group of substances (i.e., pain relievers, stimulants, sedatives, or tranquilizers). Starting with the 2007 NSDUH, a coding change was made to ensure the initiation-of-use date for the summary variable was used in conjunction with the past year recency variable for the specific substance first initiated (i.e., ANLYR). Prior to this change, the past year substance summary recency variable (i.e., PSYYR) was used in conjunction with the summary variable initiation-of-use date. For substance summary past year recency variables (i.e., PSYYR), past year use is determined if any of the substances in the group were used in the past year; thus, a small number of cases were considered past year users, but the first substance initiated among the group was not one of the substances used in the past year. A detailed example of this issue is included on page J-3 of this appendix. Given that the number of respondents affected was small, the 2007 and later year initiation variables are considered comparable with the 2002-2006 variables.
Source:	SAMHSA, Center for Behavioral Health Statistics and Quality (formerly Office of Applied Studies), National Survey on Drug Use and Health, 2002 and later years.
More detailed information about the estimates of past year initiation, possible biases, and their relevance can be found in Section B.4.1 of Appendix B of the 2013 summary of national findings (CBHSQ, 2014b).
Past Year Initiate Drug of First Use
Once past year initiates (past year incidence) have been defined, further examination can provide information on which drug was initiated first out of a group of selected drugs. Among past year initiates of illicit drugs, the first illicit drug used is determined by looking at the self-reported year and month of first use and an imputed day of first use.[footnoteRef:4] An earliest date of first use is then determined across all substances of interest. Using the earliest date of first use, a drug-of-first-use recode is created for each specific drug and the three summary measures, where recent initiates whose earliest date of first use is the same as the date of first use for the substance of interest are defined as XXXFSTYR=1 and recent initiates who did not first initiate with the specific substance are defined as XXXFSTYR=2. Respondents who are not recent initiates are defined as XXXFSTYR=3. Because respondents could initiate multiple drugs on the date of earliest use and because subcategory drugs (i.e., crack) have the same start date as main category drugs (i.e., cocaine), these measures are not mutually exclusive. [4:  See footnote 1.] 

The past year initiate drug of first use example C at the end of this appendix contains SAS code to illustrate how to create the variables used to compute estimates for the first drug initiated. The example shows a combination of four drugs: marijuana, heroin, inhalants, and stimulants. If an analyst is interested in determining the first drug initiated among a larger group of drugs, references to drugs of interest can be added to the SAS code. In addition to the SAS code, SUDAAN code is also provided to show how to estimate the number of past year initiates of marijuana, heroin, inhalants, and stimulants whose first illicit drug used out of the select group of four substances was the specific substance of interest, as well as the rate of initiation. SAS code showing how to invoke the NSDUH suppression criteria on the SUDAAN output file is also included in the example.


J-20
Example A: Past Year Initiation

/****************************************************************************
EXAMPLE OF SAS CODE TO COMPUTE ANALYSIS VARIABLES FOR PAST YEAR INITIATION

The following example SAS code creates analysis variables used in the generation of estimates for past year initiation of cigarettes, marijuana, and illicit drugs. Two separate variables are created for each substance; the first variable, ELIGXXXY, denotes whether the respondent was eligible to initiate substance use in the past year, and the second variable, RECXXXY, denotes whether the respondent initiated substance use in the past year. For both variables, XXX indicates the substance abbreviation (CIG = cigarettes, MJ or MRJ = marijuana, and SUM = illicit drug) and Y indicates which version of code is used to create the variable (2 = code for specific substance measures and 3 = code for summary substance measures only). These two variables are used in combination to estimate the rate of initiation in the past year. ****************************************************************************/

data INDICAT;   /*output dataset to be used in SUDAAN procedure*/
 set DATANAME;  /*input dataset containing NSDUH data*/

/************************************************************************ 
SAS code is included below to determine the date of initiation for the following substance summary measures: illicit drugs (SUM), illicit drugs other than marijuana (IEM), and non-medical use of psychotherapeutics (PSY). 

For substance summary measures, the date of initiation is defined as the earliest date of initiation among a group of substances. The group of substances for each summary measure is defined as follows:

Parent	Child 
 SUM	   COC,HER,HAL,INH,ANL,STM,SED,TRN,MJ
 IEM  	COC,HER,HAL,INH,ANL,STM,SED,TRN
 PSY		ANL,STM,SED,TRN

This SAS code identifies which specific substance (child) date of first use (ZZZYFU, ZZZMFU, ZZZDFU) corresponds to the substance summary measure (parent) date of first use (XXXYFU, XXXMFU, XXXDFU) and creates a check variable (XXXCKYR) that indicates the past year recency use (XXXYR)for the specific substance (child) with the earliest initiation date.
   
The check variable (XXXCKYR) will be used to create the eligibility and initiation variables for the summary measures rather than summary past year recency variables (e.g., SUMYR, IEMYR, PSYYR2). Note: A check variable was added to the 2007 NSDUH SAS code to ensure the past year recency use indicator (used to create the eligibility and initiation variables) corresponds to the substance with the earliest date of initiation among the group of substances. Please see the 2007 analytic codebook incidence appendix for details on this change.  

Note that the example code below only creates the illicit drug (SUM) summary measure, and does not create illicit drugs other than marijuana (IEM) or nonmedical use of psychotherapeutics (PSY).
*************************************************************************/

(continued)
Example A: Past Year Initiation (continued)

/*************************************************************************
    CREATION OF PAST YEAR ILLICIT DRUG CHECK VARIABLE (SUMCKYR)
*************************************************************************/

*** Define arrays for month, day and year of first use, past year recency and date of first use, and initialize illicit drug check variable ***;

array SUM_MFU (*) IRCOCMFU IRHERMFU IRHALMFU IRINHMFU IRANLMFU IRSTMMFU 
                  IRSEDMFU IRTRNMFU IRMJMFU; 

array SUM_DFU (*) IRCOCDFU IRHERDFU IRHALDFU IRINHDFU IRANLDFU IRSTMDFU 
                  IRSEDDFU IRTRNDFU IRMJDFU;

array SUM_YFU (*) IRCOCYFU IRHERYFU IRHALYFU IRINHYFU IRANLYFU IRSTMYFU  
                  IRSEDYFU IRTRNYFU IRMJYFU;
 
array SUM_YR (*) COCYR HERYR HALYR INHYR ANLYR STMYR SEDYR TRQYR MRJYR;

array F_DATE (*) COCFDATE HERFDATE HALFDATE INHFDATE ANLFDATE STMFDATE    
                 SEDFDATE TRNFDATE MJFDATE;
SUMCKYR = 0;

/* Identify which child substance date of first use corresponds to parent substance date of first use and create a past year recency check variable, XXXCKYR for the parent substance that corresponds to that particular child substance past year recency, XXXYR. If the year of first use is less than 9000 (i.e., the respondent is a lifetime user) then the date of first use needs to be considered to determine if the respondent was eligible to initiate in the past year. This check variable is only created for the three summary measures*/

*** create SAS-stored date for illicit drug summary measure from SUMMFU, SUMDFU, and SUMYFU ***;

if SUMYFU < 9000 then SUMFDATE = mdy(SUMMFU, SUMDFU, SUMYFU);  

*** create variables for each specific drug equal to SAS-stored date of 
      XXXMFU, XXXDFU, and XXXYFU ***;

do i = 1 to dim(SUM_MFU);
 if SUM_YFU(I) < 9000 then do;
  
  /* Create a first date of use variable for each drug that will be used to   create the check variables. The F_DATE variables (XXXFDATE) are only temporary variables used for this specific check.*/ 

  F_DATE(i) = mdy(SUM_MFU{i}, SUM_DFU{i}, SUM_YFU{i}); 
 
  *** set check variable (SUMCKYR) equal to specific drug year
      recency variable that corresponds to earliest first use ***;
 
  if SUMFDATE = F_DATE(i) and (SUMCKYR ne 1) then SUMCKYR = SUM_YR(i); 

 end; end;

(continued)

Example A: Past Year Initiation (continued)

/************************************************************************
Define arrays for month, day, and year of first use, past year recency, and the past year initiation variables: 
		
REC:
	RECCIG2 – Recent Initiate of Cigarette Use
	RECMJ2  – Recent Initiate of Marijuana Use
	RECSUM3 – Recent Initiate of Illicit Drug Use

ELIG:
	ELIGCIG2 – Eligible for Past Year Initiation of Cigarette Use
	ELIGMJ2  – Eligible for Past Year Initiation of Marijuana Use
	ELIGSUM3 – Eligible for Past Year Initiation of Illicit Drug Use

MFU:
	IRCIGMFU – Imputation Revised Cigarette Month of First Use
	IRMJMFU  – Imputation Revised Marijuana Month of First Use
	SUMMFU   – Any Illicit Drug Month of First Use

DFU:
	IRCIGDFU – Imputation Revised Cigarette Day of First Use
	IRMJDFU  – Imputation Revised Marijuana Day of First Use
	SUMDFU   – Any Illicit Drug Day of First Use

YFU:
	IRCIGYFU – Imputation Revised Cigarette Year of First Use
	IRMJYFU  – Imputation Revised Marijuana Year of First Use
	SUMYFU   – Any Illicit Drug Year of First Use

PY:
	CIGYR    – Cigarettes – Past Year Use
	MRJYR    – Marijuana – Past Year Use
   SUMCKYR  – Illicit Drug - Check variable indicates past year use of a
           specific illicit drug with earliest initiation date 

************************************************************************/

array REC  (*) RECCIG2 RECMJ2 RECSUM3;
array ELIG (*) ELIGCIG2 ELIGMJ2 ELIGSUM3;
array MFU  (*) IRCIGMFU IRMJMFU SUMMFU;
array DFU  (*) IRCIGDFU IRMJDFU SUMDFU;
array YFU  (*) IRCIGYFU IRMJYFU SUMYFU;
array PY   (*) CIGYR MRJYR SUMCKYR;

do I = 1 to 3;


(continued)

Example A: Past Year Initiation (continued)
/*Create the ELIGXXXY variables indicating whether a respondent was eligible for past year initiation*/

/*If the year of first use is greater than or equal to 9000 (i.e., respondent is a lifetime non-user) then they were eligible to be an initiate in the past year*/
		
if YFU(I) >= 9000 then ELIG(I) = 1;

/*If the year of first use is less than 9000 (i.e., the respondent is a lifetime user) then the date of first use needs to be considered to determine if the respondent was eligible to initiate in the past year*/
		
else if YFU(I) < 9000 then do;

/*If the difference between the interview date and the imputed date of first use is less than or equal to 365 and the respondent reported past year use, then the respondent initiated in the past year, meaning that the respondent was eligible to initiate in the past year. Note: The requirement that a respondent self-report past year use (with the exception of daily cigarette use) to be a past year initiate was first added to the SAS code for the 2005 NSDUH. Please see the 2005 analytic codebook incidence appendix for details on this change */

if (INTDATE - mdy(MFU{I}, DFU{I}, YFU{I}) <= 365) 
and PY(I) = 1 then ELIG(I) = 1;

/*Otherwise the respondent initiated at some period prior to the past year and would then be unable to initiate in the past year*/
			
else ELIG(I) = 2;
		end;


(continued)

Example A: Past Year Initiation (continued)

/*Create the RECXXXY variables indicating whether a respondent was a past year initiate or not*/
      
/*Initialize all respondents as having not been a past year initiate*/    
     
 REC(I) = 0;

/*If the respondent has a year of first use less than 9000 (i.e., the respondent is a lifetime user) then the date of first use is examined*/
     
 if YFU(I) < 9000 then do;

/*If the difference between the interview date and the imputed date of first use is less than or equal to 365 and the respondent reported past year use, then the respondent initiated in the past year */

if (INTDATE - mdy(MFU{I}, DFU{I}, YFU{I}) <= 365) and
PY(I) = 1 then REC(I) = 1;
	end;
end;
run; 


(continued)

Example A: Past Year Initiation (continued)

/****************************************************************************
EXAMPLE OF SUDAAN CODE

The following SUDAAN code illustrates how to use the variables created with the SAS code above in order to estimate the initiation rate of marijuana in the past year. The input variables include the analysis weight (ANALWT), the design-based nesting variables (VESTR and VEREP), and two of the variables created in the above SAS code (ELIGMJ2 and RECMJ2). The output includes the estimated rate of initiation (and its respective standard error) and the estimated number of initiates (and its respective standard error).

NOTE:	Estimates for only a single substance are produced with the program below. Estimates for multiple substances can be calculated simultaneously; however, keep in mind that all substance estimates will be calculated over all subgroups, and as a result, many estimates in the output will not necessarily make sense. For example, if estimates are calculated for marijuana and cigarettes, the output will show the number of initiates of marijuana among those eligible to be initiates of cigarettes.
****************************************************************************/

PROC DESCRIPT DATA=INDICAT FILETYPE=SAS DESIGN=WR;
WEIGHT ANALWT; 
NEST VESTR VEREP;
VAR RECMJ2;
SUBGROUP ELIGMJ2;
/*ONLY THE FIRST LEVEL OF ELIGMJ2 MATTERS SINCE WE ARE ONLY
  CONCERNED WITH RESPONDENTS ELIGIBLE TO BE INITIATES*/
LEVELS 1; 
SETENV COLWIDTH=12 DECWIDTH=4;
PRINT MEAN SEMEAN TOTAL SETOTAL NSUM DEFFMEAN /STYLE=NCHS;
OUTPUT MEAN SEMEAN TOTAL SETOTAL NSUM DEFFMEAN /REPLACE   
       MEANFMT=F15.10 SEMEANFMT=F15.10 
TOTALFMT=F12.0 SETOTALFMT=F12.0 
NSUMFMT=F8.0   DEFFMEANFMT=F15.10
       FILENAME="PYINIT_OUT"; RTITLE "MARIJUANA—ESTIMATES OF PAST YEAR INITIATION";   
RUN;



(continued)

Example A: Past Year Initiation (continued)

/****************************************************************************
EXAMPLE OF SAS CODE TO INVOKE THE SUPPRESSION CRITERIA

The following example SAS code invokes the suppression criteria on the SUDAAN output file from above. The variable OUTMEAN denotes the estimated rate of initiation, whereas OUTTOTAL denotes the estimated number of initiates in thousands. The corresponding standard errors are denoted by the variables OUTSEMEAN and OUTSETOTAL, respectively. For all four of these variables, an estimate that fails to meet the suppression criteria will appear as an asterisk (*).
****************************************************************************/
data PYINIT;
	set PYINIT_OUT; 

	/*calculate relative standard error (RSE)*/

	if MEAN gt 0 then RSE = SEMEAN / MEAN;
	else RSE = .;

/*calculate the relative standard error of natural log p*/

if 0 < MEAN =< 0.5 then RSELNP = RSE / abs(log(MEAN));
else if 0.5 < MEAN < 1.0 then 
RSELNP = RSE * (MEAN / (1 - MEAN)) / (abs(log(1 - MEAN)));

/*calculate effective sample size*/

EFFNSUM = NSUM / DEFFMEAN;

/*create SUPPRESS variable that will be 1 when one of the suppression   rules is met. Otherwise, SUPPRESS will be 0. Note that the variables
MEAN, RSELNP, and EFFSUM are not rounded. So, for example, if EFFNSUM=67.9999999999, the result would be suppression.*/

	if MEAN < .00005 or MEAN >= .99995 then SUPPRESS=1;
	else if RSELNP > 0.175 then SUPPRESS=1;
	else if EFFNSUM < 68 then SUPPRESS = 1;
	else if NSUM < 100 then SUPPRESS = 1;
	else SUPPRESS = 0;

	/*create OUTMEAN, OUTSEMEAN, OUTTOTAL, and OUTSETOTAL*/

	OUTMEAN = put(MEAN * 100, 10.1);
OUTSEMEAN = put(SEMEAN * 100, 10.2);
OUTTOTAL = put(TOTAL / 1000, comma12.);
OUTSETOTAL = put(SETOTAL / 1000, comma12.);

/*use the SUPPRESS variable to replace any estimates that fail the suppression criteria with an asterisk*/

	if SUPPRESS = 1 then do;
		OUTMEAN = '*';
		OUTSEMEAN = '*';
		OUTTOTAL = '*';
		OUTSETOTAL = '*';
	end;  run;
Example B: Mean Age of First Use among Past Year Initiates Aged 12 to 49

/****************************************************************************
EXAMPLE OF SUDAAN CODE

The following SUDAAN code shows how to use imputed age-of-first-use variables in order to estimate the mean age of first marijuana use among past year marijuana initiates aged 12 to 49. The input variables include the analysis weight (ANALWT), the design-based nesting variables (VESTR and VEREP), the imputed age-of-first-marijuana-use variable (IRMJAGE), an age recode (INCAGE3) that is restricted to those aged 12 to 49, and the recent initiate recode for marijuana (RECMJ2). The output includes the estimated mean age of first use (and its respective standard error).

NOTE:	Estimates for only a single substance are produced with the program below. Estimates for multiple substances can be calculated simultaneously. Further restrictions could be added if interested in those respondents who initiated prior to a specific age.
****************************************************************************/

PROC DESCRIPT	DATA=DATANAME FILETYPE=SAS DESIGN=WR MEANS DEFT;
WEIGHT ANALWT;
NEST VESTR VEREP / MISSUNIT;
VAR IRMJAGE;
SUBPOPN INCAGE3=1 AND RECMJ2=1;
SETENV COLWIDTH=12 DECWIDTH=4;
PRINT MEAN SEMEAN NSUM DEFFMEAN / STYLE=NCHS;
OUTPUT MEAN SEMEAN NSUM DEFFMEAN / REPLACE
	MEANFMT=F15.10 SEMEANFMT=F15.10
NSUMFMT=F8.0 DEFFMEANFMT=F15.10
FILENAME=“IRMJAGE_OUT”; 
RTITLE “MARIJUANA- ESTIMATES OF AGE OF FIRST USE AMONG PAST YEAR MARIJUANA INITIATES AGED 12 to 49”;
RUN; 


(continued)



Example B: Mean Age of First Use among Past Year Initiates Aged 12 to 49 (continued)

/****************************************************************************
EXAMPLE OF SAS CODE TO INVOKE THE SUPPRESSION CRITERIA

The following example SAS code invokes the suppression criteria on the SUDAAN output file from above. This differs from the suppression criteria invoking found in the other examples because this applies to averages. Estimates of means not bounded between 0 and 1 (e.g., mean age of first use, mean number of drinks consumed) were suppressed if the relative standard errors (RSEs) of the estimates were larger than .5 or if the sample sizes were smaller than 10 respondents. The variable OUTMEAN denotes the estimated mean age of first use, and the corresponding standard error is denoted by the variable OUTSEMEAN. Suppression rules are only evaluated on the means and percentages. For these two variables, an estimate that fails to meet the suppression criteria will appear as an asterisk (*).
****************************************************************************/

data IRMJAGE;
	set IRMJAGE_OUT;

	/*calculate relative standard error (RSE)*/

	if MEAN gt 0 then RSE = SEMEAN / MEAN;
	else RSE = .;

/* CALCULATE SUPPRESSION RULE FOR AVERAGES/MEAN AGE/CONT.VARS */

	IF (RSE GT 0.5) OR (NSUM < 10) THEN SUPPRESS=1;
	ELSE SUPPRESS=0;
	/*create OUTMEAN, OUTSEMEAN, */

	OUTMEAN = put(MEAN, 10.1);
OUTSEMEAN = put(SEMEAN, 10.2);
/*use the SUPPRESS variable to replace any estimates that fail the suppression criteria with an asterisk*/

	if SUPPRESS = 1 then do;
		OUTMEAN = '*';
		OUTSEMEAN = '*';
	
	end;  run;



Example C: Past Year Initiate Drug of First Use

/****************************************************************************
EXAMPLE OF SAS CODE TO COMPUTE ANALYSIS VARIABLES FOR PAST YEAR INITIATE DRUG OF FIRST USE

The following example SAS code creates analysis variables used in the generation of estimates for the first drug initiated in the past year. Included below is a group of four drugs to determine which of the four was used first. Additional drugs of interest can be added to the existing code. If interest is in all illicit drugs, the data files contain the analysis variables for each of the illicit drugs. For each substance, the variable XXXFSTYR denotes whether the past year initiate used the specific substance first, where XXX indicates the substance abbreviation. These variables are not mutually exclusive because a respondent could initiate multiple drugs on the same day as the first day they used drugs and because of child drugs having the same start date as parent drugs. ****************************************************************************/

data INDICAT;   /*output dataset to be used in SUDAAN procedure*/
	set DATANAME;  /*input dataset containing NSDUH data*/

/************************************************************************
Define arrays for recent initiates, month, day, and year of first use, and first drug used variables: 

REC:
	RECMJ2  - Recent Initiate of Marijuana Use
	RECHER2 - Recent Initiate of Heroin Use
	RECINH2 - Recent Initiate of Inhalants Use
	RECSED2 - Recent Initiate of Sedative Use

MFU:
	IRMJMFU  - Imputation Revised Marijuana Month of First Use
	IRHERMFU - Imputation Revised Heroin Month of First Use
	IRINHMFU - Imputation Revised Inhalant Month of First Use
	IRSEDMFU - Imputation Revised Sedative Month of First Use

DFU:
	IRMJDFU  - Imputation Revised Marijuana Day of First Use
	IRHERDFU - Imputation Revised Heroin Day of First Use
	IRINHDFU - Imputation Revised Inhalant Day of First Use
	IRSEDDFU - Imputation Revised Sedative Day of First Use

YFU:
	IRMJYFU  - Imputation Revised Marijuana Year of First Use
	IRHERYFU - Imputation Revised Heroin Year of First Use
	IRINHYFU - Imputation Revised Inhalant Year of First Use
	IRSEDYFU - Imputation Revised Sedative Year of First Use

FSTYR:
	MRJFSTYR - Past Year Initiate Used Marijuana First
	HERFSTYR - Past Year Initiate Used Heroin First
	INHFSTYR - Past Year Initiate Used Inhalants First
	SEDFSTYR - Past Year Initiate Used Sedatives First
************************************************************************/

(continued)

array REC       (*) RECMJ2   RECHER2  RECINH2  RECSED2;
array MFU       (*) IRMJMFU  IRHERMFU IRINHMFU IRSEDMFU;
array DFU       (*) IRMJDFU  IRHERDFU IRINHDFU IRSEDDFU;
array YFU       (*) IRMJYFU  IRHERYFU IRINHYFU IRSEDYFU;
array DATEVAR   (*) MRJDATE  HERDATE  INHDATE  SEDDATE;
array FIRSTVARS (*) MRJFSTYR HERFSTYR INHFSTYR SEDFSTYR;

/*create recency variable for group of drugs of interest*/

if sum(of REC(*)) > 0 then RECGRP = 1;
else if sum(of REC(*)) = 0 then RECGRP = 0;

/*create variables (XXXDATE) for each specific drug equal to SAS-stored date of IRXXXMFU, IRXXXDFU, and IRXXXYFU*/

do I = 1 to dim(DATEVAR);

	/*If the year of first use is less than 9000 (i.e., the respondent is a lifetime user), RECGRP = 1 (i.e., the respondent is a past year initiate of one of the drugs in the group of interest, and RECXXX = 1 (i.e., the respondent is a past year initiate of the specific drugs) then the date of use is eligible to be the earliest date of use*/

if YFU(I) < 9000 and RECGRP = 1 and REC(I) = 1 then 
DATEVAR(I) = mdy(MFU(I), DFU(I), YFU(I));

/*Otherwise, the date of use is not eligible to be the earliest date of use and XXXDATE is missing*/

	else DATEVAR(I) = .;

end;

/*** Create Earliest Date Variable using group of drugs of interest***/

MINDATE = min(MRJDATE, HERDATE, INHDATE, SEDDATE);

/****************************************************************************
   DRUG OF FIRST USE RECODES
****************************************************************************/

do I = 1 to dim(FIRSTVARS);

/*If respondent is not a recent initiate of any of the drugs in the group of interest (RECGRP = 0) then XXXFSTYR = 3*/

if RECGRP = 0 then FIRSTVARS(I) = 3;

/*If earliest date of use is the same date as the first date of use for the specific substance, then XXXFSTYR = 1*/

else if MINDATE = DATEVAR(I) then FIRSTVARS(I) = 1;

(continued)

Example C: Past Year Initiate Drug of First Use (continued)


/*Otherwise, the respondent is a recent initiate who did not initiate with the specific substance. Note that if a respondent is past year initiate who has never used the specific substance, XXXFSTYR=2*/

else FIRSTVARS(I) = 2;

end;

run; 

/****************************************************************************
EXAMPLE OF SUDAAN CODE

The following SUDAAN code illustrates how to use the variables created with the SAS code above to estimate the rate at which marijuana was the first drug initiated in the past year among past year initiates of the four drugs in the group of interest (marijuana, heroin, inhalants, and sedatives). The input variables include the analysis weight (ANALWT), the design-based nesting variables (VESTR and VEREP), and one of the variables created in the above SAS code (MRJFSTYR). The data is subset to only include recent initiates of the four drugs in the group of interest (RECGRP=1). The output includes the estimated rate of marijuana, heroin, inhalant, or sedative initiation, where the first drug initiated was marijuana (and its respective standard error) and the estimated number of marijuana, heroin, inhalant, or sedative initiates whose first drug initiated was marijuana (and its respective standard error).

NOTE:	Estimates for only a single substance are produced with the program below. Estimates for multiple substances can be calculated simultaneously.
*****************************************************************************/

PROC DESCRIPT DATA=INDICAT FILETYPE=SAS DESIGN=WR;
WEIGHT ANALWT; 
NEST VESTR VEREP;
VAR MRJFSTYR;
CATLEVEL 1;
SUBPOPN RECGRP = 1;
SETENV COLWIDTH=12 DECWIDTH=4;
PRINT PERCENT SEPERCENT TOTAL SETOTAL NSUM DEFFPCT /STYLE=NCHS;
OUTPUT PERCENT SEPERCENT TOTAL SETOTAL NSUM DEFFPCT /REPLACE   
       PERCENTFMT=F15.10 SEPERCENTFMT=F15.10
       TOTALFMT=F12.0    SETOTALFMT=F12.0 
NSUMFMT=F8.0      DEFFPCTFMT=F15.10
       FILENAME="FIRSTDRUG_OUT";
RTITLE "MARIJUANA—ESTIMATES OF FIRST DRUG INITIATED IN THE PAST YEAR";   
RUN;

/****************************************************************************
EXAMPLE OF SAS CODE TO INVOKE THE SUPPRESSION CRITERIA

The following example SAS code invokes the suppression criteria on the SUDAAN output file. The variable OUTMEAN denotes the estimated rate of marijuana, heroin, inhalant, or sedative initiation, where the first drug initiated was marijuana, whereas OUTTOTAL denotes the estimated number in thousands of marijuana, heroin, inhalant, or sedative initiates whose first drug initiated was marijuana. The corresponding standard errors are denoted by the variables OUTSEMEAN and OUTSETOTAL, respectively. For all four of these variables, an estimate that fails to meet the suppression criteria will appear as an asterisk (*).
****************************************************************************/

data FIRSTDRUG;
	set FIRSTDRUG_OUT; 

	/*Calculate the mean from PERCENT and SEPERCENT output from SUDAAN*/

	MEAN = PERCENT / 100;
	SEMEAN = SEPERCENT / 100;
	DEFFMEAN = DEFFPCT;

	/*calculate relative standard error (RSE)*/

	if MEAN > 0 then RSE = SEMEAN / MEAN;
	else RSE = .;

/*calculate the relative standard error of natural log p*/

if 0 < MEAN =< 0.5 then RSELNP = RSE / abs(log(MEAN));
else if 0.5 < MEAN < 1.0 then 
    RSELNP = RSE * (MEAN / (1 - MEAN)) / (abs(log(1 - MEAN)));

/*calculate effective sample size*/

EFFNSUM = NSUM / DEFFMEAN;

/*create SUPPRESS variable that will be 1 when one of the suppression   rules is met. Otherwise, SUPPRESS will be 0. Note that the variables
MEAN, RSELNP, and EFFSUM are not rounded. So, for example, if     EFFNSUM=67.9999999999, the result would be suppression.*/

	if MEAN < .00005 or MEAN >= .99995 then SUPPRESS=1;
	else if RSELNP > 0.175 then SUPPRESS=1;
	else if EFFNSUM < 68 then SUPPRESS = 1;
	else if NSUM < 100 then SUPPRESS = 1;
	else SUPPRESS = 0;

	

(continued)

Example C: Past Year Initiate Drug of First Use (continued)


/*create OUTMEAN, OUTSEMEAN, OUTTOTAL, and OUTSETOTAL*/

	OUTMEAN = put(MEAN * 100, 10.1);
OUTSEMEAN = put(SEMEAN * 100, 10.2);
OUTTOTAL = put(TOTAL / 1000, comma12.);
OUTSETOTAL = put(SETOTAL / 1000, comma12.);

/*use the SUPPRESS variable to replace any estimates that fail the suppression criteria with an asterisk*/

	if SUPPRESS = 1 then do;
		OUTMEAN = '*';
		OUTSEMEAN = '*';
		OUTTOTAL = '*';
		OUTSETOTAL = '*';
	end;
run;



References
Center for Behavioral Health Statistics and Quality. (2014a). Results from the 2013 National Survey on Drug Use and Health: Detailed tables. Rockville, MD: Substance Abuse and Mental Health Services Administration.
Center for Behavioral Health Statistics and Quality. (2014b). Results from the 2013 National Survey on Drug Use and Health: Summary of national findings (HHS Publication No. SMA 14-4863, NSDUH Series H-48). Rockville, MD: Substance Abuse and Mental Health Services Administration.
Gfroerer, J., Hughes, A., Chromy, J., Heller, D., & Packer, L. (2004, July). Estimating trends in substance use based on reports of prior use in a cross-sectional survey. In S. B. Cohen & J. M. Lepkowski (Eds.), Eighth Conference on Health Survey Research Methods: Conference proceedings [Peachtree City, GA] (DHHS Publication No. PHS 04-1013, pp. 29-34). Hyattsville, MD: National Center for Health Statistics. [Available as a PDF at https://www.cdc.gov/nchs/products/hsrmc.htm]
RTI International. (2012). SUDAAN® language manual, Release 11.0.0. Research Triangle Park, NC: RTI International.
SAS Institute Inc. (2013). SAS/STAT® 13.1 user's guide. Cary, NC: SAS Institute Inc.
